

★ライブテキスト作成★

採掘ゲームを作るには

1

初期設定・キャラの表示

```
10 / *MINING  
20 CLY:CLS  
30 X=15:Y=11  
40 LC X,Y  
50 ?CHR$(249)
```



3

キャラの移動

```
10 / *MINING
20 CLV:CLS
30 X=15:Y=11
40 LC X,Y
50 ?CHR$(249)
```

```
100 @LOOP
110 U=X-BTN(28)+BTN(29)
120 V=Y-BTN(30)+BTN(31)
130 LC X,Y: ?CHR$(0);
140 LC U,V: ?CHR$(249);
150 X=U:Y=V
160 GOTO @LOOP
```



4

WAITを入れる

```
10 / *MINING
20 CLV:CLS
30 X=15:Y=11
40 LC X,Y
50 ?CHR$(249)
```

```
100 @LOOP
110 U=X-BTN(28)+BTN(29)
120 V=Y-BTN(30)+BTN(31)
130 LC X,Y: ?CHR$(0);
140 LC U,V: ?CHR$(249);
150 X=U:Y=V
155 WAIT 5
160 GOTO @LOOP
```



5

はみだしチェック

```
100 @LOOP
110 U=X-BTN(28)+BTN(29)
112 IF U<0 OR U>31 GOTO @LOOP
120 V=Y-BTN(30)+BTN(31)
122 IF V<0 OR V>22 GOTO @LOOP
130 LC X,Y:?CHR$(0);
140 LC U,V:?CHR$(249);
150 X=U:Y=V
155 WAIT 5
160 GOTO @LOOP
```

キャラが画面を
はみ出さなくなる

6

キーを押していない時は動かさない

```
100 @LOOP
110 U=X-BTN(28)+BTN(29)
112 IF U<0 OR U>31 GOTO @LOOP
120 V=Y-BTN(30)+BTN(31)
122 IF V<0 OR V>22 GOTO @LOOP
124 IF X=U AND Y=V GOTO @LOOP
130 LC X,Y:?CHR$(0);
140 LC U,V:?CHR$(249);
150 X=U:Y=V
155 WAIT 5
160 GOTO @LOOP
```


キャラが
点滅しなくなる

7

障害物を出す

```
10 / *MINING
20 CLV:CLS
25 FOR I=1 TO 50
26 LC RND(32),RND(23)
27 ?CHR$(1);
28 NEXT
30 X=15:Y=11
40 LC X,Y
:
```

画面に障害物が出る




8

障害物チェック

```
:
110 U=X-BTN(28)+BTN(29)
112 IF U<0 OR U>31 GOTO @LOOP
120 V=Y-BTN(30)+BTN(31)
122 IF V<0 OR V>22 GOTO @LOOP
125 C=SCR(U,V)
128 IF C>0 GOTO @LOOP
:
```

障害物に当たると
それ以上動けない



9

採掘をする(1)

```
:  
100 @LOOP  
110 U=X-BTN(28)+BTN(29)  
112 IF U<0 OR U>31 GOTO @LOOP  
120 V=Y-BTN(30)+BTN(31)  
122 IF V<0 OR V>22 GOTO @LOOP  
125 C=SCR(U,V)  
126 IF BTN(32) GSB @ACT:GOTO @LOOP  
128 IF C>0 GOTO @LOOP  
:
```

スペースキーが押されていたら
アクションのサブルーチンを呼ぶ

10

採掘をする(2)

```
:  
200 @ACT  
210 GSB @GET  
220 RTN  
230 @GET  
240 LC U,V:?"o";  
250 WAIT 30  
260 H=42+U+V  
270 LC U,V:?"CHR$(H);  
280 WAIT 30  
290 LC U,V:?"CHR$(0);  
300 LC 0,23:?"(" ;CHR$(H);")";  
310 WAIT 30  
320 RTN
```



採掘して
手に持った物を表示

スペースキーを押しながら
矢印キーを押すと採掘できる

11

採掘の制限

```
:  
200 @ACT  
205 IF C=1 RTN  
210 IF H=0 GSB @GET  
220 RTN
```

- 行き先が障害物の時は採掘できないので戻る
- 手に何も持っていない時だけ採掘する



12

持っている物を置く

```
200 @ACT  
205 IF C=1 RTN  
210 IF H=0 GSB @GET ELSE GSB @PUT  
220 RTN
```

手に何も持っていなかったら採掘
持っていたらそれを置く

```
:  
330 @PUT  
340 LC U,V: ?CHR$(H);  
350 LC 0,23: ?" ( )";  
360 WAIT 30  
370 H=0:RTN
```

スペース


持っている物を置くサブルーチン



13

物を持つ

```
230 @GET  
235 IF C>0 H=C:GOTO @TAKE  
240 LC U,V:?"o";  
250 WAIT 30  
260 H=42+U+V  
270 LC U,V:?"CHR$(H);"  
280 WAIT 30  
285 @TAKE  
290 LC U,V:?"CHR$(0);"  
300 LC 0,23:?"( ";CHR$(H);" )";  
310 WAIT 30  
320 RTN
```




何も持っていなかったら
スペース+矢印キーで
置いてある物を持つ

14

物を置いて合成する

```
330 @PUT  
340 LC U,V:?"CHR$(H);"  
350 LC 0,23:?"( )";  
360 WAIT 30  
370 IF C=0 H=0:RTN  
380 LC U,V:?"+";  
390 WAIT 30  
400 H=(C+H)&255  
410 IF H<33 H=H+33  
420 IF H=127 OR H=128 H=129  
430 LC U,V:?"CHR$(H);"  
440 WAIT 30  
450 H=0:RTN
```



既に物がある所に物を置くと
合成して新しい物を作る

15

効果音を出す

250 BEEP 30:WAIT 30	穴を掘る音
310 BEEP:WAIT 30	物をゲットした音
360 BEEP 30:WAIT 30	物を置く音
440 BEEP 10,30:WAIT 30	物を合成した音

16

採掘ゲーム・完成

```
10 /*MINING タイトル
20 CLV:CLS 変数クリア:画面クリア
25 FOR I=1 TO 50
26 LC RND(32),RND(23) } 障害物をランダムな位置に
27 ?CHR$(1); } 50個表示
28 NEXT
30 X=15:Y=11 キャラのX,Y座標セット
40 LC X,Y }
50 ?CHR$(249) } キャラを表示
100 @LOOP
110 U=X-BTN(28)+BTN(29)
112 IF U<0 OR U>31 GOTO @LOOP
120 V=Y-BTN(30)+BTN(31)
122 IF V<0 OR V>22 GOTO @LOOP
124 IF X=U AND Y=V GOTO @LOOP
125 C=SCR(U,V) 移動先にある物を読み取り
```

17

```

126 IF BTN(32) GSB @ACT:GOTO @LOOP
128 IF C>0 GOTO @LOOP 障害物チェック
130 LC X,Y:?CHR$(0);
140 LC U,V:?CHR$(249);
150 X=U:Y=V
155 WAIT 5 時間待ち
160 GOTO @LOOP 移動のくりかえし
200 @ACT スペースキーアクションのサブルーチン
205 IF C=1 RTN 行き先が障害物だったらもどる
210 IF H=0 GSB @GET ELSE GSB @PUT
220 RTN 物を持っていなかったら@GET、持っていたら@PUTを呼ぶ
230 @GET 採掘サブルーチン
235 IF C>0 H=C:GOTO @TAKE 行き先に物があったら拾う
240 LC U,V:? "o";
250 BEEP 30:WAIT 30
260 H=42+U+V
270 LC U,V:?CHR$(H);
280 WAIT 30

```

18

```

285 @TAKE
290 LC U,V:?CHR$(0);
300 LC 0,23:? "(";CHR$(H);")";
310 BEEP:WAIT 30
320 RTN
330 @PUT 物を置くサブルーチン
340 LC U,V:?CHR$(H);
350 LC 0,23:? "( )";
360 BEEP 30:WAIT 30
370 IF C=0 H=0:RTN 移動先に物が無かったらもどる
380 LC U,V:? "+";
390 WAIT 30
400 H=(C+H)&255
410 IF H<33 H=H+33
420 IF H=127 OR H=128 H=129
430 LC U,V:?CHR$(H);
440 BEEP 10,30:WAIT 30
450 H=0:RTN 手持ちをクリアしてもどる

```

19